Rules for integrands of the form $u\,(a+b\,\mathrm{ArcSin}[c\,x])^n$

1. $\int (d+e\,x)^m\,(a+b\,\mathrm{ArcSin}[c\,x])^n\,dx$

   1. $\int (d+e\,x)^m\,(a+b\,\mathrm{ArcSin}[c\,x])^n\,dx$ when $n\in\mathbb{Z}^+$

      1: $\int \frac{(a+b\,\mathrm{ArcSin}[c\,x])^n}{d+e\,x}\,dx$ when $n\in\mathbb{Z}^+$

- Derivation: Integration by substitution
- Basis: $\frac{1}{d+e\,x} == \mathrm{Subst}\left[\frac{\mathrm{Cos}[x]}{c\,d+e\,\mathrm{Sin}[x]},\,x,\,\mathrm{ArcSin}[c\,x]\right]\,\partial_x \mathrm{ArcSin}[c\,x]$
- Note: $\frac{(a+b\,x)^n\,\mathrm{Cos}[x]}{c\,d+e\,\mathrm{Sin}[x]}$ is not integrable unless $n\in\mathbb{Z}^+$.
- Rule: If $n\in\mathbb{Z}^+$, then

$$\int \frac{(a+b\,\mathrm{ArcSin}[c\,x])^n}{d+e\,x}\,dx \longrightarrow \mathrm{Subst}\left[\int \frac{(a+b\,x)^n\,\mathrm{Cos}[x]}{c\,d+e\,\mathrm{Sin}[x]}\,dx,\,x,\,\mathrm{ArcSin}[c\,x]\right]$$

- Program code:

```
Int[(a_.+b_.*ArcSin[c_.*x_])^n_./(d_+e_.*x_),x_Symbol] :=
  Subst[Int[(a+b*x)^n*Cos[x]/(c*d+e*Sin[x]),x],x,ArcSin[c*x]] /;
FreeQ[{a,b,c,d,e},x] && IGtQ[n,0]
```

```
Int[(a_.+b_.*ArcCos[c_.*x_])^n_./(d_+e_.*x_),x_Symbol] :=
  -Subst[Int[(a+b*x)^n*Sin[x]/(c*d+e*Cos[x]),x],x,ArcCos[c*x]] /;
FreeQ[{a,b,c,d,e},x] && IGtQ[n,0]
```

2: $\int (d + e\,x)^m\,(a + b\,\text{ArcSin}[c\,x])^n\,dx$ when $n \in \mathbb{Z}^+ \wedge m \neq -1$

- **Reference: G&R 2.831, CRC 453, A&S 4.4.65**
- **Reference: G&R 2.832, CRC 454, A&S 4.4.67**
- **Derivation: Integration by parts**
- **Basis: If** $m \neq -1$**, then** $(d + e\,x)^m = \partial_x \frac{(d+e\,x)^{m+1}}{e\,(m+1)}$
- **Rule: If** $n \in \mathbb{Z}^+ \wedge m \neq -1$**, then**

$$\int (d + e\,x)^m\,(a + b\,\text{ArcSin}[c\,x])^n\,dx \longrightarrow \frac{(d + e\,x)^{m+1}\,(a + b\,\text{ArcSin}[c\,x])^n}{e\,(m + 1)} - \frac{b\,c\,n}{e\,(m + 1)} \int \frac{(d + e\,x)^{m+1}\,(a + b\,\text{ArcSin}[c\,x])^{n-1}}{\sqrt{1 - c^2\,x^2}}\,dx$$

- **Program code:**

```
Int[(d_+e_.*x_)^m_.*(a_.+b_.*ArcSin[c_.*x_])^n_.,x_Symbol] :=
  (d+e*x)^(m+1)*(a+b*ArcSin[c*x])^n/(e*(m+1)) -
  b*c*n/(e*(m+1))*Int[(d+e*x)^(m+1)*(a+b*ArcSin[c*x])^(n-1)/Sqrt[1-c^2*x^2],x] /;
FreeQ[{a,b,c,d,e,m},x] && IGtQ[n,0] && NeQ[m,-1]
```

```
Int[(d_+e_.*x_)^m_.*(a_.+b_.*ArcCos[c_.*x_])^n_.,x_Symbol] :=
  (d+e*x)^(m+1)*(a+b*ArcCos[c*x])^n/(e*(m+1)) +
  b*c*n/(e*(m+1))*Int[(d+e*x)^(m+1)*(a+b*ArcCos[c*x])^(n-1)/Sqrt[1-c^2*x^2],x] /;
FreeQ[{a,b,c,d,e,m},x] && IGtQ[n,0] && NeQ[m,-1]
```

2. $\int (d + e x)^m (a + b \text{ArcSin}[c x])^n \, dx$ when $m \in \mathbb{Z}^+$

1: $\int (d + e x)^m (a + b \text{ArcSin}[c x])^n \, dx$ when $m \in \mathbb{Z}^+ \land n < -1$

Derivation: Algebraic expansion

Rule: If $m \in \mathbb{Z}^+ \land n < -1$**, then**

$$\int (d + e x)^m (a + b \text{ArcSin}[c x])^n \, dx \longrightarrow \int \text{ExpandIntegrand}[(d + e x)^m (a + b \text{ArcSin}[c x])^n, x] \, dx$$

Program code:

```
Int[(d_+e_.*x_)^m_.*(a_.+b_.*ArcSin[c_.*x_])^n_,x_Symbol] :=
  Int[ExpandIntegrand[(d+e*x)^m*(a+b*ArcSin[c*x])^n,x],x] /;
FreeQ[{a,b,c,d,e},x] && IGtQ[m,0] && LtQ[n,-1]
```

```
Int[(d_+e_.*x_)^m_.*(a_.+b_.*ArcCos[c_.*x_])^n_,x_Symbol] :=
  Int[ExpandIntegrand[(d+e*x)^m*(a+b*ArcCos[c*x])^n,x],x] /;
FreeQ[{a,b,c,d,e},x] && IGtQ[m,0] && LtQ[n,-1]
```

2: $\int (d + e x)^m (a + b \text{ArcSin}[c x])^n \, dx$ when $m \in \mathbb{Z}^+$

Derivation: Integration by substitution

Basis: $F[x] == \frac{1}{c} F\left[\frac{\text{Sin}[\text{ArcSin}[c x]]}{c}\right] \text{Cos}[\text{ArcSin}[c x]] \, \partial_x \text{ArcSin}[c x]$

Note: If $m \in \mathbb{Z}^+$**, then** $(a + b x)^n \text{Cos}[x] (c d + e \text{Sin}[x])^m$ **is integrable in closed-form.**

Rule: If $m \in \mathbb{Z}^+$**, then**

$$\int (d + e x)^m (a + b \text{ArcSin}[c x])^n \, dx \longrightarrow \frac{1}{c^{m+1}} \text{Subst}\left[\int (a + b x)^n \text{Cos}[x] (c d + e \text{Sin}[x])^m \, dx, x, \text{ArcSin}[c x]\right]$$

Program code:

```
Int[(d_.+e_.*x_)^m_.*(a_.+b_.*ArcSin[c_.*x_])^n_,x_Symbol] :=
  1/c^(m+1)*Subst[Int[(a+b*x)^n*Cos[x]*(c*d+e*Sin[x])^m,x],x,ArcSin[c*x]] /;
FreeQ[{a,b,c,d,e,n},x] && IGtQ[m,0]
```

```
Int[(d_.+e_.*x_)^m_.*(a_.+b_.*ArcCos[c_.*x_])^n_,x_Symbol] :=
  -1/c^(m+1)*Subst[Int[(a+b*x)^n*Sin[x]*(c*d+e*Cos[x])^m,x],x,ArcCos[c*x]] /;
FreeQ[{a,b,c,d,e,n},x] && IGtQ[m,0]
```

2. $\int P_x\,(a+b\,\text{ArcSin}[c\,x])^n\,dx$

1: $\int P_x\,(a+b\,\text{ArcSin}[c\,x])\,dx$

Derivation: Integration by parts

Rule: Let $u=\int P_x\,dx$**, then**

$$\int P_x\,(a+b\,\text{ArcSin}[c\,x])\,dx \longrightarrow u\,(a+b\,\text{ArcSin}[c\,x])-b\,c\int\frac{u}{\sqrt{1-c^2\,x^2}}\,dx$$

Program code:

```
Int[Px_*(a_.+b_.*ArcSin[c_.*x_]),x_Symbol] :=
  With[{u=IntHide[ExpandExpression[Px,x],x]},
  Dist[a+b*ArcSin[c*x],u,x] - b*c*Int[SimplifyIntegrand[u/Sqrt[1-c^2*x^2],x],x]] /;
FreeQ[{a,b,c},x] && PolynomialQ[Px,x]
```

```
Int[Px_*(a_.+b_.*ArcCos[c_.*x_]),x_Symbol] :=
  With[{u=IntHide[ExpandExpression[Px,x],x]},
  Dist[a+b*ArcCos[c*x],u,x] + b*c*Int[SimplifyIntegrand[u/Sqrt[1-c^2*x^2],x],x]] /;
FreeQ[{a,b,c},x] && PolynomialQ[Px,x]
```

x: $\int P_x (a + b \operatorname{ArcSin}[c x])^n \, dx$ when $n \in \mathbb{Z}^+$

Derivation: Integration by parts

Rule: If $n \in \mathbb{Z}^+$, let $u = \int P_x \, dx$, then

$$\int P_x (a + b \operatorname{ArcSin}[c x])^n \, dx \longrightarrow u (a + b \operatorname{ArcSin}[c x])^n - b c n \int \frac{u (a + b \operatorname{ArcSin}[c x])^{n-1}}{\sqrt{1 - c^2 x^2}} \, dx$$

Program code:

```
(* Int[Px_*(a_.+b_.*ArcSin[c_.*x_])^n_.,x_Symbol] :=
  With[{u=IntHide[Px,x]},
  Dist[(a+b*ArcSin[c*x])^n,u,x] - b*c*n*Int[SimplifyIntegrand[u*(a+b*ArcSin[c*x])^(n-1)/Sqrt[1-c^2*x^2],x],x]] /;
FreeQ[{a,b,c},x] && PolynomialQ[Px,x] && IGtQ[n,0] *)
```

```
(* Int[Px_*(a_.+b_.*ArcCos[c_.*x_])^n_.,x_Symbol] :=
  With[{u=IntHide[Px,x]},
  Dist[(a+b*ArcCos[c*x])^n,u,x] + b*c*n*Int[SimplifyIntegrand[u*(a+b*ArcCos[c*x])^(n-1)/Sqrt[1-c^2*x^2],x],x]] /;
FreeQ[{a,b,c},x] && PolynomialQ[Px,x] && IGtQ[n,0] *)
```

2: $\int P_x (a + b \operatorname{ArcSin}[c x])^n \, dx$ when $n \neq 1$

Derivation: Algebraic expansion

Rule: If $n \neq 1$, then

$$\int P_x (a + b \operatorname{ArcSin}[c x])^n \, dx \longrightarrow \int \operatorname{ExpandIntegrand}[P_x (a + b \operatorname{ArcSin}[c x])^n, x] \, dx$$

Program code:

```
Int[Px_*(a_.+b_.*ArcSin[c_.*x_])^n_,x_Symbol] :=
  Int[ExpandIntegrand[Px*(a+b*ArcSin[c*x])^n,x],x] /;
FreeQ[{a,b,c,n},x] && PolynomialQ[Px,x]
```

```
Int[Px_*(a_.+b_.*ArcCos[c_.*x_])^n_,x_Symbol] :=
  Int[ExpandIntegrand[Px*(a+b*ArcCos[c*x])^n,x],x] /;
FreeQ[{a,b,c,n},x] && PolynomialQ[Px,x]
```

3. $\int P_x (d+e x)^m (a+b \operatorname{ArcSin}[c x])^n \, dx$ when $n \in \mathbb{Z}^+$

1: $\int P_x (d+e x)^m (a+b \operatorname{ArcSin}[c x]) \, dx$

Derivation: Integration by parts

Rule: Let $u = \int P_x (d+e x)^m \, dx$**, then**

$$\int P_x (d+e x)^m (a+b \operatorname{ArcSin}[c x]) \, dx \longrightarrow u (a+b \operatorname{ArcSin}[c x]) - b c \int \frac{u}{\sqrt{1-c^2 x^2}} \, dx$$

Program code:

```
Int[Px_*(d_.+e_.*x_)^m_.*(a_.+b_.*ArcSin[c_.*x_]),x_Symbol] :=
  With[{u=IntHide[Px*(d+e*x)^m,x]},
  Dist[a+b*ArcSin[c*x],u,x] - b*c*Int[SimplifyIntegrand[u/Sqrt[1-c^2*x^2],x],x]] /;
FreeQ[{a,b,c,d,e,m},x] && PolynomialQ[Px,x]
```

```
Int[Px_*(d_.+e_.*x_)^m_.*(a_.+b_.*ArcCos[c_.*x_]),x_Symbol] :=
  With[{u=IntHide[Px*(d+e*x)^m,x]},
  Dist[a+b*ArcCos[c*x],u,x] + b*c*Int[SimplifyIntegrand[u/Sqrt[1-c^2*x^2],x],x]] /;
FreeQ[{a,b,c,d,e,m},x] && PolynomialQ[Px,x]
```

2: $\int (f+g x)^p (d+e x)^m (a+b \operatorname{ArcSin}[c x])^n \, dx$ when $(n \mid p) \in \mathbb{Z}^+ \wedge m \in \mathbb{Z}^- \wedge m+p+1<0$

Derivation: Integration by parts

Note: If $p \in \mathbb{Z}^+ \wedge m \in \mathbb{Z}^- \wedge m+p+1<0$**, then** $\int (f+g x)^p (d+e x)^m \, dx$ **is a rational function.**

Rule: If $(n \mid p) \in \mathbb{Z}^+ \wedge m \in \mathbb{Z}^- \wedge m+p+1<0$**, let** $u = \int (f+g x)^p (d+e x)^m \, dx$**, then**

$$\int (f+g x)^p (d+e x)^m (a+b \operatorname{ArcSin}[c x])^n \, dx \longrightarrow u (a+b \operatorname{ArcSin}[c x])^n - b c n \int \frac{u (a+b \operatorname{ArcSin}[c x])^{n-1}}{\sqrt{1-c^2 x^2}} \, dx$$

Program code:

```
Int[(f_.+g_.*x_)^p_.*(d_+e_.*x_)^m_*(a_.+b_.*ArcSin[c_.*x_])^n_,x_Symbol] :=
  With[{u=IntHide[(f+g*x)^p*(d+e*x)^m,x]},
  Dist[(a+b*ArcSin[c*x])^n,u,x] - b*c*n*Int[SimplifyIntegrand[u*(a+b*ArcSin[c*x])^(n-1)/Sqrt[1-c^2*x^2],x],x]] /;
FreeQ[{a,b,c,d,e,f,g},x] && IGtQ[n,0] && IGtQ[p,0] && ILtQ[m,0] && LtQ[m+p+1,0]
```

```
Int[(f_.+g_.*x_)^p_.*(d_+e_.*x_)^m_*(a_.+b_.*ArcCos[c_.*x_])^n_,x_Symbol] :=
  With[{u=IntHide[(f+g*x)^p*(d+e*x)^m,x]},
  Dist[(a+b*ArcCos[c*x])^n,u,x] + b*c*n*Int[SimplifyIntegrand[u*(a+b*ArcCos[c*x])^(n-1)/Sqrt[1-c^2*x^2],x],x]] /;
FreeQ[{a,b,c,d,e,f,g},x] && IGtQ[n,0] && IGtQ[p,0] && ILtQ[m,0] && LtQ[m+p+1,0]
```

3: $\int \frac{(f+g x+h x^2)^p (a+b \operatorname{ArcSin}[c x])^n}{(d+e x)^2} \, dx$ when $(n \mid p) \in \mathbb{Z}^+ \wedge e g - 2 d h == 0$

Derivation: Integration by parts

- **Note: If $p \in \mathbb{Z}^+ \wedge e g - 2 d h == 0$, then $\int \frac{(f+g x+h x^2)^p}{(d+e x)^2} \, dx$ is a rational function.**

- **Rule: If $(n \mid p) \in \mathbb{Z}^+ \wedge e g - 2 d h == 0$, let $u = \int \frac{(f+g x+h x^2)^p}{(d+e x)^2} \, dx$, then**

$$\int \frac{(f+g x+h x^2)^p (a+b \operatorname{ArcSin}[c x])^n}{(d+e x)^2} \, dx \longrightarrow u (a+b \operatorname{ArcSin}[c x])^n - b c n \int \frac{u (a+b \operatorname{ArcSin}[c x])^{n-1}}{\sqrt{1-c^2 x^2}} \, dx$$

Program code:

```
Int[(f_.+g_.*x_+h_.*x_^2)^p_.*(a_.+b_.*ArcSin[c_.*x_])^n_/(d_+e_.*x_)^2,x_Symbol] :=
  With[{u=IntHide[(f+g*x+h*x^2)^p/(d+e*x)^2,x]},
  Dist[(a+b*ArcSin[c*x])^n,u,x] - b*c*n*Int[SimplifyIntegrand[u*(a+b*ArcSin[c*x])^(n-1)/Sqrt[1-c^2*x^2],x],x]] /;
FreeQ[{a,b,c,d,e,f,g,h},x] && IGtQ[n,0] && IGtQ[p,0] && EqQ[e*g-2*d*h,0]
```

```
Int[(f_.+g_.*x_+h_.*x_^2)^p_.*(a_.+b_.*ArcCos[c_.*x_])^n_/(d_+e_.*x_)^2,x_Symbol] :=
  With[{u=IntHide[(f+g*x+h*x^2)^p/(d+e*x)^2,x]},
  Dist[(a+b*ArcCos[c*x])^n,u,x] + b*c*n*Int[SimplifyIntegrand[u*(a+b*ArcCos[c*x])^(n-1)/Sqrt[1-c^2*x^2],x],x]] /;
FreeQ[{a,b,c,d,e,f,g,h},x] && IGtQ[n,0] && IGtQ[p,0] && EqQ[e*g-2*d*h,0]
```

4: $\int P_x (d+e x)^m (a+b \text{ArcSin}[c x])^n \, dx$ when $n \in \mathbb{Z}^+ \land m \in \mathbb{Z}$

Derivation: Algebraic expansion

Rule: If $n \in \mathbb{Z}^+ \land m \in \mathbb{Z}$**, then**

$$\int P_x (d+e x)^m (a+b \text{ArcSin}[c x])^n \, dx \longrightarrow \int \text{ExpandIntegrand}[P_x (d+e x)^m (a+b \text{ArcSin}[c x])^n, x] \, dx$$

Program code:

```
Int[Px_*(d_+e_.*x_)^m_.*(a_.+b_.*ArcSin[c_.*x_])^n_,x_Symbol] :=
  Int[ExpandIntegrand[Px*(d+e*x)^m*(a+b*ArcSin[c*x])^n,x],x] /;
FreeQ[{a,b,c,d,e},x] && PolynomialQ[Px,x] && IGtQ[n,0] && IntegerQ[m]
```

```
Int[Px_*(d_+e_.*x_)^m_.*(a_.+b_.*ArcCos[c_.*x_])^n_,x_Symbol] :=
  Int[ExpandIntegrand[Px*(d+e*x)^m*(a+b*ArcCos[c*x])^n,x],x] /;
FreeQ[{a,b,c,d,e},x] && PolynomialQ[Px,x] && IGtQ[n,0] && IntegerQ[m]
```

4. $\int (f+g x)^m \left(d+e x^2\right)^p (a+b \text{ArcSin}[c x])^n \, dx$ when $c^2 d+e=0 \land m \in \mathbb{Z} \land p-\frac{1}{2} \in \mathbb{Z}$

1. $\int (f+g x)^m \left(d+e x^2\right)^p (a+b \text{ArcSin}[c x])^n \, dx$ when $c^2 d+e=0 \land m \in \mathbb{Z} \land p-\frac{1}{2} \in \mathbb{Z} \land d>0$

1: $\int (f+g x)^m \left(d+e x^2\right)^p (a+b \text{ArcSin}[c x]) \, dx$ when $c^2 d+e=0 \land m \in \mathbb{Z}^+ \land p+\frac{1}{2} \in \mathbb{Z}^- \land d>0 \land (m<-2 p-1 \lor m>3)$

Derivation: Integration by parts

- **Note: If** $m \in \mathbb{Z} \land p+\frac{1}{2} \in \mathbb{Z} \land 0<m<-2 p-1$**, then** $\int (f+g x)^m \left(d+e x^2\right)^p dx$ **is an algebraic function.**
- **Rule: If** $c^2 d+e=0 \land m \in \mathbb{Z}^+ \land p+\frac{1}{2} \in \mathbb{Z}^- \land d>0 \land (m<-2 p-1 \lor m>3)$**, let** $u=\int (f+g x)^m \left(d+e x^2\right)^p dx$**, then**

$$\int (f+g x)^m \left(d+e x^2\right)^p (a+b \text{ArcSin}[c x]) \, dx \longrightarrow u (a+b \text{ArcSin}[c x]) - b c \int \frac{u}{\sqrt{1-c^2 x^2}} \, dx$$

Program code:

```
Int[(f_+g_.*x_)^m_.*(d_+e_.*x_^2)^p_*(a_.+b_.*ArcSin[c_.*x_]),x_Symbol] :=
  With[{u=IntHide[(f+g*x)^m*(d+e*x^2)^p,x]},
  Dist[a+b*ArcSin[c*x],u,x] - b*c*Int[Dist[1/Sqrt[1-c^2*x^2],u,x],x]] /;
FreeQ[{a,b,c,d,e,f,g},x] && EqQ[c^2*d+e,0] && IGtQ[m,0] && ILtQ[p+1/2,0] && GtQ[d,0] && (LtQ[m,-2*p-1] || GtQ[m,3])
```

```
Int[(f_+g_.*x_)^m_.*(d_+e_.*x_^2)^p_*(a_.+b_.*ArcCos[c_.*x_]),x_Symbol] :=
  With[{u=IntHide[(f+g*x)^m*(d+e*x^2)^p,x]},
  Dist[a+b*ArcCos[c*x],u,x] + b*c*Int[Dist[1/Sqrt[1-c^2*x^2],u,x],x]] /;
FreeQ[{a,b,c,d,e,f,g},x] && EqQ[c^2*d+e,0] && IGtQ[m,0] && ILtQ[p+1/2,0] && GtQ[d,0] && (LtQ[m,-2*p-1] || GtQ[m,3])
```

2: $\int (f+g x)^m \left(d+e x^2\right)^p (a+b \operatorname{ArcSin}[c x])^n \, dx$ when

$c^2 d+e=0 \bigwedge m\in\mathbb{Z}^+ \bigwedge p+\frac{1}{2}\in\mathbb{Z} \bigwedge d>0 \bigwedge n\in\mathbb{Z}^+ \bigwedge (m=1 \bigvee p>0 \bigvee (n=1 \bigwedge p>-1) \bigvee (m=2 \bigwedge p<-2))$

Derivation: Algebraic expansion

Rule: If $c^2 d+e=0 \bigwedge m\in\mathbb{Z} \bigwedge p+\frac{1}{2}\in\mathbb{Z} \bigwedge d>0 \bigwedge n\in\mathbb{Z}^+ \bigwedge m>0$**, then**

$$\int (f+g x)^m \left(d+e x^2\right)^p (a+b \operatorname{ArcSin}[c x])^n \, dx \longrightarrow \int \left(d+e x^2\right)^p (a+b \operatorname{ArcSin}[c x])^n \operatorname{ExpandIntegrand}[(f+g x)^m, x] \, dx$$

Program code:

```
Int[(f_+g_.*x_)^m_.*(d_+e_.*x_^2)^p_*(a_.+b_.*ArcSin[c_.*x_])^n_.,x_Symbol] :=
  Int[ExpandIntegrand[(d+e*x^2)^p*(a+b*ArcSin[c*x])^n,(f+g*x)^m,x],x] /;
FreeQ[{a,b,c,d,e,f,g},x] && EqQ[c^2*d+e,0] && IGtQ[m,0] && IntegerQ[p+1/2] && GtQ[d,0] && IGtQ[n,0] &&
  (m==1 || p>0 || n==1 && p>-1 || m==2 && p<-2)
```

```
Int[(f_+g_.*x_)^m_.*(d_+e_.*x_^2)^p_*(a_.+b_.*ArcCos[c_.*x_])^n_.,x_Symbol] :=
  Int[ExpandIntegrand[(d+e*x^2)^p*(a+b*ArcCos[c*x])^n,(f+g*x)^m,x],x] /;
FreeQ[{a,b,c,d,e,f,g},x] && EqQ[c^2*d+e,0] && IGtQ[m,0] && IntegerQ[p+1/2] && GtQ[d,0] && IGtQ[n,0] &&
  (m==1 || p>0 || n==1 && p>-1 || m==2 && p<-2)
```

3. $\int (f+g x)^m \left(d+e x^2\right)^p (a+b \operatorname{ArcSin}[c x])^n \, dx$ when $c^2 d+e=0 \bigwedge m\in\mathbb{Z} \bigwedge p+\frac{1}{2}\in\mathbb{Z}^+ \bigwedge d>0$

1: $\int (f+g x)^m \sqrt{d+e x^2} \, (a+b \operatorname{ArcSin}[c x])^n \, dx$ when $c^2 d+e=0 \wedge m\in\mathbb{Z}^- \wedge d>0 \wedge n\in\mathbb{Z}^+$

Derivation: Integration by parts

Basis: If $c^2 d+e=0 \wedge d>0$**, then** $\frac{(a+b \operatorname{ArcSin}[c x])^n}{\sqrt{d+e x^2}} = \partial_x \frac{(a+b \operatorname{ArcSin}[c x])^{n+1}}{b c \sqrt{d} \, (n+1)}$

Rule: If $c^2 d+e=0 \wedge m\in\mathbb{Z}^- \wedge d>0 \wedge n\in\mathbb{Z}^+$**, then**

$$\int (f+g x)^m \sqrt{d+e x^2} \, (a+b \operatorname{ArcSin}[c x])^n \, dx \longrightarrow$$

$$\frac{(f+g x)^m \left(d+e x^2\right) (a+b \text{ArcSin}[c x])^{n+1}}{b c \sqrt{d} (n+1)}-$$
$$\frac{1}{b c \sqrt{d} (n+1)} \int \left(d g m+2 e f x+e g (m+2) x^2\right) (f+g x)^{m-1} (a+b \text{ArcSin}[c x])^{n+1} dx$$

Program code:

```
Int[(f_+g_.*x_)^m_*Sqrt[d_+e_.*x_^2]*(a_.+b_.*ArcSin[c_.*x_])^n_.,x_Symbol] :=
  (f+g*x)^m*(d+e*x^2)*(a+b*ArcSin[c*x])^(n+1)/(b*c*Sqrt[d]*(n+1)) -
  1/(b*c*Sqrt[d]*(n+1))*Int[(d*g*m+2*e*f*x+e*g*(m+2)*x^2)*(f+g*x)^(m-1)*(a+b*ArcSin[c*x])^(n+1),x] /;
FreeQ[{a,b,c,d,e,f,g},x] && EqQ[c^2*d+e,0] && ILtQ[m,0] && GtQ[d,0] && IGtQ[n,0]
```

```
Int[(f_+g_.*x_)^m_*Sqrt[d_+e_.*x_^2]*(a_.+b_.*ArcCos[c_.*x_])^n_.,x_Symbol] :=
  -(f+g*x)^m*(d+e*x^2)*(a+b*ArcCos[c*x])^(n+1)/(b*c*Sqrt[d]*(n+1)) +
  1/(b*c*Sqrt[d]*(n+1))*Int[(d*g*m+2*e*f*x+e*g*(m+2)*x^2)*(f+g*x)^(m-1)*(a+b*ArcCos[c*x])^(n+1),x] /;
FreeQ[{a,b,c,d,e,f,g},x] && EqQ[c^2*d+e,0] && ILtQ[m,0] && GtQ[d,0] && IGtQ[n,0]
```

2: $\int (f+g x)^m \left(d+e x^2\right)^p (a+b \text{ArcSin}[c x])^n dx$ when $c^2 d+e=0 \bigwedge m\in\mathbb{Z} \bigwedge p+\frac{1}{2}\in\mathbb{Z}^+ \bigwedge d>0 \bigwedge n\in\mathbb{Z}^+$

Derivation: Algebraic expansion

Rule: If $c^2 d+e=0 \bigwedge m\in\mathbb{Z} \bigwedge p+\frac{1}{2}\in\mathbb{Z}^+ \bigwedge d>0 \bigwedge n\in\mathbb{Z}^+$**, then**

$$\int (f+g x)^m \left(d+e x^2\right)^p (a+b \text{ArcSin}[c x])^n dx \longrightarrow \int \sqrt{d+e x^2} (a+b \text{ArcSin}[c x])^n \text{ExpandIntegrand}\left[(f+g x)^m \left(d+e x^2\right)^{p-1/2}, x\right] dx$$

Program code:

```
Int[(f_+g_.*x_)^m_.*(d_+e_.*x_^2)^p_*(a_.+b_.*ArcSin[c_.*x_])^n_.,x_Symbol] :=
  Int[ExpandIntegrand[Sqrt[d+e*x^2]*(a+b*ArcSin[c*x])^n,(f+g*x)^m*(d+e*x^2)^(p-1/2),x],x] /;
FreeQ[{a,b,c,d,e,f,g},x] && EqQ[c^2*d+e,0] && IntegerQ[m] && IGtQ[p+1/2,0] && GtQ[d,0] && IGtQ[n,0]
```

```
Int[(f_+g_.*x_)^m_.*(d_+e_.*x_^2)^p_*(a_.+b_.*ArcCos[c_.*x_])^n_.,x_Symbol] :=
  Int[ExpandIntegrand[Sqrt[d+e*x^2]*(a+b*ArcCos[c*x])^n,(f+g*x)^m*(d+e*x^2)^(p-1/2),x],x] /;
FreeQ[{a,b,c,d,e,f,g},x] && EqQ[c^2*d+e,0] && IntegerQ[m] && IGtQ[p+1/2,0] && GtQ[d,0] && IGtQ[n,0]
```

3: $\int (f+g x)^m \left(d+e x^2\right)^p (a+b \text{ArcSin}[c x])^n \, dx$ when $c^2 d+e==0 \bigwedge m \in \mathbb{Z}^- \bigwedge p-\frac{1}{2} \in \mathbb{Z}^+ \bigwedge d>0 \bigwedge n \in \mathbb{Z}^+$

Derivation: Integration by parts

- **Basis: If** $c^2 d+e==0 \wedge d>0$, **then** $\frac{(a+b \text{ArcSin}[c x])^n}{\sqrt{d+e x^2}} == \partial_x \frac{(a+b \text{ArcSin}[c x])^{n+1}}{b c \sqrt{d} (n+1)}$
- **Rule: If** $c^2 d+e==0 \bigwedge m \in \mathbb{Z}^- \bigwedge p-\frac{1}{2} \in \mathbb{Z}^+ \bigwedge d>0 \bigwedge n \in \mathbb{Z}^+$, **then**

$$\int (f+g x)^m \left(d+e x^2\right)^p (a+b \text{ArcSin}[c x])^n \, dx \longrightarrow$$

$$\frac{(f+g x)^m \left(d+e x^2\right)^{p+\frac{1}{2}} (a+b \text{ArcSin}[c x])^{n+1}}{b c \sqrt{d} (n+1)} -$$

$$\frac{1}{b c \sqrt{d} (n+1)} \int (f+g x)^{m-1} (a+b \text{ArcSin}[c x])^{n+1} \text{ExpandIntegrand}\left[\left(d g m+e f (2 p+1) x+e g (m+2 p+1) x^2\right) \left(d+e x^2\right)^{p-\frac{1}{2}}, x\right] dx$$

Program code:

```
Int[(f_+g_.*x_)^m_.*(d_+e_.*x_^2)^p_*(a_.+b_.*ArcSin[c_.*x_])^n_.,x_Symbol] :=
  (f+g*x)^m*(d+e*x^2)^(p+1/2)*(a+b*ArcSin[c*x])^(n+1)/(b*c*Sqrt[d]*(n+1)) -
  1/(b*c*Sqrt[d]*(n+1))*
    Int[ExpandIntegrand[(f+g*x)^(m-1)*(a+b*ArcSin[c*x])^(n+1),(d*g*m+e*f*(2*p+1)*x+e*g*(m+2*p+1)*x^2)*(d+e*x^2)^(p-1/2),x],x] /;
FreeQ[{a,b,c,d,e,f,g},x] && EqQ[c^2*d+e,0] && ILtQ[m,0] && IGtQ[p-1/2,0] && GtQ[d,0] && IGtQ[n,0]
```

```
Int[(f_+g_.*x_)^m_.*(d_+e_.*x_^2)^p_*(a_.+b_.*ArcCos[c_.*x_])^n_.,x_Symbol] :=
  -(f+g*x)^m*(d+e*x^2)^(p+1/2)*(a+b*ArcCos[c*x])^(n+1)/(b*c*Sqrt[d]*(n+1)) +
  1/(b*c*Sqrt[d]*(n+1))*
    Int[ExpandIntegrand[(f+g*x)^(m-1)*(a+b*ArcCos[c*x])^(n+1),(d*g*m+e*f*(2*p+1)*x+e*g*(m+2*p+1)*x^2)*(d+e*x^2)^(p-1/2),x],x] /;
FreeQ[{a,b,c,d,e,f,g},x] && EqQ[c^2*d+e,0] && ILtQ[m,0] && IGtQ[p-1/2,0] && GtQ[d,0] && IGtQ[n,0]
```

4. $\int (f+g x)^m \left(d+e x^2\right)^p (a+b \operatorname{ArcSin}[c x])^n \, dx$ when $c^2 d+e=0 \bigwedge m \in \mathbb{Z} \bigwedge p-\frac{1}{2} \in \mathbb{Z}^- \bigwedge d>0$

1. $\int \frac{(f+g x)^m (a+b \operatorname{ArcSin}[c x])^n}{\sqrt{d+e x^2}} \, dx$ when $c^2 d+e=0 \wedge m \in \mathbb{Z} \wedge d>0$

1: $\int \frac{(f+g x)^m (a+b \operatorname{ArcSin}[c x])^n}{\sqrt{d+e x^2}} \, dx$ when $c^2 d+e=0 \wedge m \in \mathbb{Z}^+ \wedge d>0 \wedge n<-1$

- **Derivation: Integration by parts**

- **Basis: If** $c^2 d+e=0 \wedge d>0$**, then** $\frac{(a+b \operatorname{ArcSin}[c x])^n}{\sqrt{d+e x^2}} = \partial_x \frac{(a+b \operatorname{ArcSin}[c x])^{n+1}}{b c \sqrt{d} (n+1)}$

- **Rule: If** $c^2 d+e=0 \wedge m \in \mathbb{Z} \wedge d>0 \wedge m>0 \wedge n<-1$**, then**

$$\int \frac{(f+g x)^m (a+b \operatorname{ArcSin}[c x])^n}{\sqrt{d+e x^2}} \, dx \longrightarrow \frac{(f+g x)^m (a+b \operatorname{ArcSin}[c x])^{n+1}}{b c \sqrt{d} (n+1)} - \frac{g m}{b c \sqrt{d} (n+1)} \int (f+g x)^{m-1} (a+b \operatorname{ArcSin}[c x])^{n+1} \, dx$$

- **Program code:**

```
Int[(f_+g_.*x_)^m_.*(a_.+b_.*ArcSin[c_.*x_])^n_/Sqrt[d_+e_.*x_^2],x_Symbol] :=
  (f+g*x)^m*(a+b*ArcSin[c*x])^(n+1)/(b*c*Sqrt[d]*(n+1)) -
  g*m/(b*c*Sqrt[d]*(n+1))*Int[(f+g*x)^(m-1)*(a+b*ArcSin[c*x])^(n+1),x] /;
FreeQ[{a,b,c,d,e,f,g},x] && EqQ[c^2*d+e,0] && IGtQ[m,0] && GtQ[d,0] && LtQ[n,-1]
```

```
Int[(f_+g_.*x_)^m_.*(a_.+b_.*ArcCos[c_.*x_])^n_/Sqrt[d_+e_.*x_^2],x_Symbol] :=
  -(f+g*x)^m*(a+b*ArcCos[c*x])^(n+1)/(b*c*Sqrt[d]*(n+1)) +
  g*m/(b*c*Sqrt[d]*(n+1))*Int[(f+g*x)^(m-1)*(a+b*ArcCos[c*x])^(n+1),x] /;
FreeQ[{a,b,c,d,e,f,g},x] && EqQ[c^2*d+e,0] && IGtQ[m,0] && GtQ[d,0] && LtQ[n,-1]
```

2: $\int \frac{(f+g x)^m (a+b \operatorname{ArcSin}[c x])^n}{\sqrt{d+e x^2}} \, dx$ when $c^2 d+e=0 \wedge m \in \mathbb{Z} \wedge d>0 \wedge (m>0 \vee n \in \mathbb{Z}^+)$

Derivation: Integration by substitution

Basis: If $c^2 d+e=0 \wedge d>0$**, then** $\frac{F[x]}{\sqrt{d+e x^2}} = \frac{1}{c \sqrt{d}} \operatorname{Subst}\left[F\left[\frac{\operatorname{Sin}[x]}{c}\right], x, \operatorname{ArcSin}[c x]\right] \partial_x \operatorname{ArcSin}[c x]$

Rule: If $c^2 d+e=0 \wedge m \in \mathbb{Z} \wedge d>0 \wedge (m>0 \vee n \in \mathbb{Z}^+)$**, then**

$$\int \frac{(f+g x)^m (a+b \operatorname{ArcSin}[c x])^n}{\sqrt{d+e x^2}} \, dx \to \frac{1}{c^{m+1} \sqrt{d}} \operatorname{Subst}\left[\int (a+b x)^n (c f+g \operatorname{Sin}[x])^m \, dx, x, \operatorname{ArcSin}[c x]\right]$$

Program code:

```
Int[(f_+g_.*x_)^m_.*(a_.+b_.*ArcSin[c_.*x_])^n_./Sqrt[d_+e_.*x_^2],x_Symbol] :=
  1/(c^(m+1)*Sqrt[d])*Subst[Int[(a+b*x)^n*(c*f+g*Sin[x])^m,x],x,ArcSin[c*x]] /;
FreeQ[{a,b,c,d,e,f,g,n},x] && EqQ[c^2*d+e,0] && IntegerQ[m] && GtQ[d,0] && (GtQ[m,0] || IGtQ[n,0])
```

```
Int[(f_+g_.*x_)^m_.*(a_.+b_.*ArcCos[c_.*x_])^n_./Sqrt[d_+e_.*x_^2],x_Symbol] :=
  -1/(c^(m+1)*Sqrt[d])*Subst[Int[(a+b*x)^n*(c*f+g*Cos[x])^m,x],x,ArcCos[c*x]] /;
FreeQ[{a,b,c,d,e,f,g,n},x] && EqQ[c^2*d+e,0] && IntegerQ[m] && GtQ[d,0] && (GtQ[m,0] || IGtQ[n,0])
```

2: $\int (f+g x)^m \left(d+e x^2\right)^p (a+b \operatorname{ArcSin}[c x])^n \, dx$ when $c^2 d+e=0 \wedge m \in \mathbb{Z} \wedge p+\frac{1}{2} \in \mathbb{Z}^- \wedge d>0 \wedge n \in \mathbb{Z}^+$

Derivation: Algebraic expansion

Rule: If $c^2 d+e=0 \wedge m \in \mathbb{Z} \wedge p+\frac{1}{2} \in \mathbb{Z}^- \wedge d>0 \wedge n \in \mathbb{Z}^+$**, then**

$$\int (f+g x)^m \left(d+e x^2\right)^p (a+b \operatorname{ArcSin}[c x])^n \, dx \to \int \frac{(a+b \operatorname{ArcSin}[c x])^n}{\sqrt{d+e x^2}} \operatorname{ExpandIntegrand}\left[(f+g x)^m \left(d+e x^2\right)^{p+1/2}, x\right] dx$$

Program code:

```
Int[(f_+g_.*x_)^m_.*(d_+e_.*x_^2)^p_*(a_.+b_.*ArcSin[c_.*x_])^n_.,x_Symbol] :=
  Int[ExpandIntegrand[(a+b*ArcSin[c*x])^n/Sqrt[d+e*x^2],(f+g*x)^m*(d+e*x^2)^(p+1/2),x],x] /;
FreeQ[{a,b,c,d,e,f,g},x] && EqQ[c^2*d+e,0] && IntegerQ[m] && ILtQ[p+1/2,0] && GtQ[d,0] && IGtQ[n,0]
```

```
Int[(f_+g_.*x_)^m_.*(d_+e_.*x_^2)^p_*(a_.+b_.*ArcCos[c_.*x_])^n_.,x_Symbol] :=
  Int[ExpandIntegrand[(a+b*ArcCos[c*x])^n/Sqrt[d+e*x^2],(f+g*x)^m*(d+e*x^2)^(p+1/2),x],x] /;
FreeQ[{a,b,c,d,e,f,g},x] && EqQ[c^2*d+e,0] && IntegerQ[m] && ILtQ[p+1/2,0] && GtQ[d,0] && IGtQ[n,0]
```

2: $\int (f+g\,x)^m \left(d+e\,x^2\right)^p (a+b\,\text{ArcSin}[c\,x])^n\,dx$ when $c^2 d+e=0 \wedge m\in\mathbb{Z} \wedge p-\frac{1}{2}\in\mathbb{Z} \wedge d \ngtr 0$

Derivation: Piecewise constant extraction

- **Basis: If** $c^2 d+e=0$, **then** $\partial_x \frac{(d+e\,x^2)^p}{(1-c^2 x^2)^p}=0$

- **Rule: If** $c^2 d+e=0 \wedge m\in\mathbb{Z} \wedge p-\frac{1}{2}\in\mathbb{Z} \wedge d \ngtr 0$, **then**

$$\int (f+g\,x)^m \left(d+e\,x^2\right)^p (a+b\,\text{ArcSin}[c\,x])^n\,dx \rightarrow \frac{d^{\text{IntPart}[p]} \left(d+e\,x^2\right)^{\text{FracPart}[p]}}{\left(1-c^2 x^2\right)^{\text{FracPart}[p]}} \int (f+g\,x)^m \left(1-c^2 x^2\right)^p (a+b\,\text{ArcSin}[c\,x])^n\,dx$$

Program code:

```
Int[(f_+g_.*x_)^m_.*(d_+e_.*x_^2)^p_*(a_.+b_.*ArcSin[c_.*x_])^n_.,x_Symbol] :=
  d^IntPart[p]*(d+e*x^2)^FracPart[p]/(1-c^2*x^2)^FracPart[p]*Int[(f+g*x)^m*(1-c^2*x^2)^p*(a+b*ArcSin[c*x])^n,x] /;
FreeQ[{a,b,c,d,e,f,g,n},x] && EqQ[c^2*d+e,0] && IntegerQ[m] && IntegerQ[p-1/2] && Not[GtQ[d,0]]
```

```
Int[(f_+g_.*x_)^m_.*(d_+e_.*x_^2)^p_*(a_.+b_.*ArcCos[c_.*x_])^n_.,x_Symbol] :=
  d^IntPart[p]*(d+e*x^2)^FracPart[p]/(1-c^2*x^2)^FracPart[p]*Int[(f+g*x)^m*(1-c^2*x^2)^p*(a+b*ArcCos[c*x])^n,x] /;
FreeQ[{a,b,c,d,e,f,g,n},x] && EqQ[c^2*d+e,0] && IntegerQ[m] && IntegerQ[p-1/2] && Not[GtQ[d,0]]
```

5. $\int \text{Log}[h\,(f+g\,x)^m]\,\left(d+e\,x^2\right)^p\,(a+b\,\text{ArcSin}[c\,x])^n\,dx$ when $c^2\,d+e=0\,\bigwedge\,p-\frac{1}{2}\in\mathbb{Z}$

1. $\int \text{Log}[h\,(f+g\,x)^m]\,\left(d+e\,x^2\right)^p\,(a+b\,\text{ArcSin}[c\,x])^n\,dx$ when $c^2\,d+e=0\,\bigwedge\,p-\frac{1}{2}\in\mathbb{Z}\,\bigwedge\,d>0$

1: $\int \frac{\text{Log}[h\,(f+g\,x)^m]\,(a+b\,\text{ArcSin}[c\,x])^n}{\sqrt{d+e\,x^2}}\,dx$ when $c^2\,d+e=0\,\wedge\,d>0\,\wedge\,n\in\mathbb{Z}^+$

Derivation: Integration by parts

- **Basis: If** $c^2\,d+e=0\,\wedge\,d>0$**, then** $\frac{(a+b\,\text{ArcSin}[c\,x])^n}{\sqrt{d+e\,x^2}}=\partial_x\frac{(a+b\,\text{ArcSin}[c\,x])^{n+1}}{b\,c\,\sqrt{d}\,(n+1)}$
- **Note: If** $n\in\mathbb{Z}^+$**, then** $\frac{(a+b\,\text{ArcSin}[c\,x])^{n+1}}{f+g\,x}$ **is integrable in closed-form.**
- **Rule: If** $c^2\,d+e=0\,\wedge\,d>0\,\wedge\,n\in\mathbb{Z}^+$**, then**

$$\int \frac{\text{Log}[h\,(f+g\,x)^m]\,(a+b\,\text{ArcSin}[c\,x])^n}{\sqrt{d+e\,x^2}}\,dx\,\longrightarrow\,\frac{\text{Log}[h\,(f+g\,x)^m]\,(a+b\,\text{ArcSin}[c\,x])^{n+1}}{b\,c\,\sqrt{d}\,(n+1)}-\frac{g\,m}{b\,c\,\sqrt{d}\,(n+1)}\int \frac{(a+b\,\text{ArcSin}[c\,x])^{n+1}}{f+g\,x}\,dx$$

Program code:

```
Int[Log[h_.*(f_.+g_.*x_)^m_.]*(a_.+b_.*ArcSin[c_.*x_])^n_./Sqrt[d_+e_.*x_^2],x_Symbol] :=
  Log[h*(f+g*x)^m]*(a+b*ArcSin[c*x])^(n+1)/(b*c*Sqrt[d]*(n+1)) -
  g*m/(b*c*Sqrt[d]*(n+1))*Int[(a+b*ArcSin[c*x])^(n+1)/(f+g*x),x] /;
FreeQ[{a,b,c,d,e,f,g,h,m},x] && EqQ[c^2*d+e,0] && GtQ[d,0] && IGtQ[n,0]
```

```
Int[Log[h_.*(f_.+g_.*x_)^m_.]*(a_.+b_.*ArcCos[c_.*x_])^n_./Sqrt[d_+e_.*x_^2],x_Symbol] :=
  -Log[h*(f+g*x)^m]*(a+b*ArcCos[c*x])^(n+1)/(b*c*Sqrt[d]*(n+1)) +
  g*m/(b*c*Sqrt[d]*(n+1))*Int[(a+b*ArcCos[c*x])^(n+1)/(f+g*x),x] /;
FreeQ[{a,b,c,d,e,f,g,h,m},x] && EqQ[c^2*d+e,0] && GtQ[d,0] && IGtQ[n,0]
```

2: $\int \text{Log}[h\,(f+g\,x)^m]\,\left(d+e\,x^2\right)^p\,(a+b\,\text{ArcSin}[c\,x])^n\,dx$ when $c^2\,d+e=0\,\bigwedge\,p-\frac{1}{2}\in\mathbb{Z}\,\bigwedge\,d\not>0$

Derivation: Piecewise constant extraction

- **Basis: If** $c^2\,d+e=0$**, then** $\partial_x\frac{(d+e\,x^2)^p}{(1-c^2\,x^2)^p}=0$
- **Rule: If** $c^2\,d+e=0\,\bigwedge\,p-\frac{1}{2}\in\mathbb{Z}\,\bigwedge\,d\not>0$**, then**

$$\int \text{Log}[h\,(f+g\,x)^m]\,\left(d+e\,x^2\right)^p\,(a+b\,\text{ArcSin}[c\,x])^n\,dx \longrightarrow \frac{d^{\text{IntPart}[p]}\,\left(d+e\,x^2\right)^{\text{FracPart}[p]}}{\left(1-c^2\,x^2\right)^{\text{FracPart}[p]}}\int \text{Log}[h\,(f+g\,x)^m]\,\left(1-c^2\,x^2\right)^p\,(a+b\,\text{ArcSin}[c\,x])^n\,dx$$

Program code:

```
Int[Log[h_.*(f_.+g_.*x_)^m_.]*(d_+e_.*x_^2)^p_*(a_.+b_.*ArcSin[c_.*x_])^n_.,x_Symbol] :=
  d^IntPart[p]*(d+e*x^2)^FracPart[p]/(1-c^2*x^2)^FracPart[p]*Int[Log[h*(f+g*x)^m]*(1-c^2*x^2)^p*(a+b*ArcSin[c*x])^n,x] /;
FreeQ[{a,b,c,d,e,f,g,h,m,n},x] && EqQ[c^2*d+e,0] && IntegerQ[p-1/2] && Not[GtQ[d,0]]
```

```
Int[Log[h_.*(f_.+g_.*x_)^m_.]*(d_+e_.*x_^2)^p_*(a_.+b_.*ArcCos[c_.*x_])^n_.,x_Symbol] :=
  d^IntPart[p]*(d+e*x^2)^FracPart[p]/(1-c^2*x^2)^FracPart[p]*Int[Log[h*(f+g*x)^m]*(1-c^2*x^2)^p*(a+b*ArcCos[c*x])^n,x] /;
FreeQ[{a,b,c,d,e,f,g,h,m,n},x] && EqQ[c^2*d+e,0] && IntegerQ[p-1/2] && Not[GtQ[d,0]]
```

6. $\int (d+e\,x)^m\,(f+g\,x)^m\,(a+b\,\text{ArcSin}[c\,x])^n\,dx$

1: $\int (d+e\,x)^m\,(f+g\,x)^m\,(a+b\,\text{ArcSin}[c\,x])\,dx$ when $m+\frac{1}{2}\in\mathbb{Z}^-$

Derivation: Integration by parts

Rule: If $m+\frac{1}{2}\in\mathbb{Z}^-$**, let** $u=\int(d+e\,x)^m\,(f+g\,x)^m\,dx$**, then**

$$\int (d+e\,x)^m\,(f+g\,x)^m\,(a+b\,\text{ArcSin}[c\,x])\,dx \longrightarrow u\,(a+b\,\text{ArcSin}[c\,x])-b\,c\int\frac{u}{\sqrt{1-c^2\,x^2}}\,dx$$

Program code:

```
Int[(d_+e_.*x_)^m_*(f_+g_.*x_)^m_*(a_.+b_.*ArcSin[c_.*x_]),x_Symbol] :=
  With[{u=IntHide[(d+e*x)^m*(f+g*x)^m,x]},
  Dist[a+b*ArcSin[c*x],u,x] - b*c*Int[Dist[1/Sqrt[1-c^2*x^2],u,x],x]] /;
FreeQ[{a,b,c,d,e,f,g},x] && ILtQ[m+1/2,0]
```

```
Int[(d_+e_.*x_)^m_*(f_+g_.*x_)^m_*(a_.+b_.*ArcCos[c_.*x_]),x_Symbol] :=
  With[{u=IntHide[(d+e*x)^m*(f+g*x)^m,x]},
  Dist[a+b*ArcCos[c*x],u,x] + b*c*Int[Dist[1/Sqrt[1-c^2*x^2],u,x],x]] /;
FreeQ[{a,b,c,d,e,f,g},x] && ILtQ[m+1/2,0]
```

2: $\int (d+e\,x)^m\,(f+g\,x)^m\,(a+b\,\text{ArcSin}[c\,x])^n\,dx$ when $m\in\mathbb{Z}$

Derivation: Algebraic expansion

Rule: If $m\in\mathbb{Z}$, then

$$\int (d+e\,x)^m\,(f+g\,x)^m\,(a+b\,\text{ArcSin}[c\,x])^n\,dx\;\longrightarrow\;\int \text{ExpandIntegrand}\left[(d+e\,x)^m\,(f+g\,x)^m\,(a+b\,\text{ArcSin}[c\,x])^n,\;x\right]dx$$

Program code:

```
Int[(d_+e_.*x_)^m_.*(f_+g_.*x_)^m_.*(a_.+b_.*ArcSin[c_.*x_])^n_.,x_Symbol] :=
  Int[ExpandIntegrand[(d+e*x)^m*(f+g*x)^m*(a+b*ArcSin[c*x])^n,x],x] /;
FreeQ[{a,b,c,d,e,f,g,n},x] && IntegerQ[m]
```

```
Int[(d_+e_.*x_)^m_.*(f_+g_.*x_)^m_.*(a_.+b_.*ArcCos[c_.*x_])^n_.,x_Symbol] :=
  Int[ExpandIntegrand[(d+e*x)^m*(f+g*x)^m*(a+b*ArcCos[c*x])^n,x],x] /;
FreeQ[{a,b,c,d,e,f,g,n},x] && IntegerQ[m]
```

7: $\int u\,(a+b\,\text{ArcSin}[c\,x])\,dx$ when $\int u\,dx$ is free of inverse functions

Derivation: Integration by parts

Rule: Let $v = \int u\,dx$, if v is free of inverse functions, then

$$\int u\,(a+b\,\text{ArcSin}[c\,x])\,dx\;\longrightarrow\;v\,(a+b\,\text{ArcSin}[c\,x])-b\,c\int\frac{v}{\sqrt{1-c^2\,x^2}}\,dx$$

Program code:

```
Int[u_*(a_.+b_.*ArcSin[c_.*x_]),x_Symbol] :=
  With[{v=IntHide[u,x]},
  Dist[a+b*ArcSin[c*x],v,x] - b*c*Int[SimplifyIntegrand[v/Sqrt[1-c^2*x^2],x],x] /;
 InverseFunctionFreeQ[v,x]] /;
FreeQ[{a,b,c},x]
```

```
Int[u_*(a_.+b_.*ArcCos[c_.*x_]),x_Symbol] :=
  With[{v=IntHide[u,x]},
  Dist[a+b*ArcCos[c*x],v,x] + b*c*Int[SimplifyIntegrand[v/Sqrt[1-c^2*x^2],x],x] /;
 InverseFunctionFreeQ[v,x]] /;
FreeQ[{a,b,c},x]
```

8. $\int P_x\, u\, (a+b\,\text{ArcSin}[c\,x])^n\, dx$

1: $\int P_x\, (d+e\,x^2)^p\, (a+b\,\text{ArcSin}[c\,x])^n\, dx$ when $c^2\,d+e=0\, \bigwedge\, p-\frac{1}{2}\in\mathbb{Z}$

Derivation: Algebraic expansion

Rule: If $c^2\,d+e=0\, \bigwedge\, p-\frac{1}{2}\in\mathbb{Z}$, then

$$\int P_x\, (d+e\,x^2)^p\, (a+b\,\text{ArcSin}[c\,x])^n\, dx \,\longrightarrow\, \int \text{ExpandIntegrand}\left[P_x\, (d+e\,x^2)^p\, (a+b\,\text{ArcSin}[c\,x])^n,\, x\right] dx$$

Program code:

```
Int[Px_*(d_+e_.*x_^2)^p_*(a_.+b_.*ArcSin[c_.*x_])^n_.,x_Symbol] :=
  With[{u=ExpandIntegrand[Px*(d+e*x^2)^p*(a+b*ArcSin[c*x])^n,x]},
  Int[u,x] /;
 SumQ[u]] /;
FreeQ[{a,b,c,d,e,n},x] && PolynomialQ[Px,x] && EqQ[c^2*d+e,0] && IntegerQ[p-1/2]
```

```
Int[Px_*(d_+e_.*x_^2)^p_*(a_.+b_.*ArcCos[c_.*x_])^n_.,x_Symbol] :=
  With[{u=ExpandIntegrand[Px*(d+e*x^2)^p*(a+b*ArcCos[c*x])^n,x]},
  Int[u,x] /;
 SumQ[u]] /;
FreeQ[{a,b,c,d,e,n},x] && PolynomialQ[Px,x] && EqQ[c^2*d+e,0] && IntegerQ[p-1/2]
```

2: $\int P_x\, \left(f+g\,(d+e\,x^2)^p\right)^m\, (a+b\,\text{ArcSin}[c\,x])^n\, dx$ when $c^2\,d+e=0\, \bigwedge\, p+\frac{1}{2}\in\mathbb{Z}^+\, \bigwedge\, (m\mid n)\in\mathbb{Z}$

Derivation: Algebraic expansion

Rule: If $c^2\,d+e=0\, \bigwedge\, p+\frac{1}{2}\in\mathbb{Z}^+\, \bigwedge\, (m\mid n)\in\mathbb{Z}$, then

$$\int P_x\, \left(f+g\,(d+e\,x^2)^p\right)^m\, (a+b\,\text{ArcSin}[c\,x])^n\, dx \,\longrightarrow\, \int \text{ExpandIntegrand}\left[P_x\, \left(f+g\,(d+e\,x^2)^p\right)^m\, (a+b\,\text{ArcSin}[c\,x])^n,\, x\right] dx$$

Program code:

```
Int[Px_.*(f_+g_.*(d_+e_.*x_^2)^p_)^m_.*(a_.+b_.*ArcSin[c_.*x_])^n_.,x_Symbol] :=
  With[{u=ExpandIntegrand[Px*(f+g*(d+e*x^2)^p)^m*(a+b*ArcSin[c*x])^n,x]},
  Int[u,x] /;
 SumQ[u]] /;
FreeQ[{a,b,c,d,e,f,g},x] && PolynomialQ[Px,x] && EqQ[c^2*d+e,0] && IGtQ[p+1/2,0] && IntegersQ[m,n]
```

```
Int[Px_.*(f_+g_.*(d_+e_.*x_^2)^p_)^m_.*(a_.+b_.*ArcCos[c_.*x_])^n_.,x_Symbol] :=
  With[{u=ExpandIntegrand[Px*(f+g*(d+e*x^2)^p)^m*(a+b*ArcCos[c*x])^n,x]},
  Int[u,x] /;
 SumQ[u]] /;
FreeQ[{a,b,c,d,e,f,g},x] && PolynomialQ[Px,x] && EqQ[c^2*d+e,0] && IGtQ[p+1/2,0] && IntegersQ[m,n]
```

9. $\int RF_x\, u\, (a+b\, \text{ArcSin}[c\, x])^n\, dx$ when $n \in \mathbb{Z}^+$

1. $\int RF_x\, (a+b\, \text{ArcSin}[c\, x])^n\, dx$ when $n \in \mathbb{Z}^+$

1: $\int RF_x\, \text{ArcSin}[c\, x]^n\, dx$ when $n \in \mathbb{Z}^+$

Derivation: Algebraic expansion

Rule: If $n \in \mathbb{Z}^+$, **then**

$$\int RF_x\, \text{ArcSin}[c\, x]^n\, dx \longrightarrow \int \text{ArcSin}[c\, x]^n\, \text{ExpandIntegrand}[RF_x,\, x]\, dx$$

Program code:

```
Int[RFx_*ArcSin[c_.*x_]^n_.,x_Symbol] :=
  With[{u=ExpandIntegrand[ArcSin[c*x]^n,RFx,x]},
  Int[u,x] /;
 SumQ[u]] /;
FreeQ[c,x] && RationalFunctionQ[RFx,x] && IGtQ[n,0]
```

```
Int[RFx_*ArcCos[c_.*x_]^n_.,x_Symbol] :=
  With[{u=ExpandIntegrand[ArcCos[c*x]^n,RFx,x]},
  Int[u,x] /;
 SumQ[u]] /;
FreeQ[c,x] && RationalFunctionQ[RFx,x] && IGtQ[n,0]
```

2: $\int RF_x\,(a+b\,\text{ArcSin}[c\,x])^n\,dx$ when $n\in\mathbb{Z}^+$

Derivation: Algebraic expansion

Rule: If $n\in\mathbb{Z}^+$, then

$$\int RF_x\,(a+b\,\text{ArcSin}[c\,x])^n\,dx\;\longrightarrow\;\int \text{ExpandIntegrand}[RF_x\,(a+b\,\text{ArcSin}[c\,x])^n,\,x]\,dx$$

Program code:

```
Int[RFx_*(a_+b_.*ArcSin[c_.*x_])^n_.,x_Symbol] :=
  Int[ExpandIntegrand[RFx*(a+b*ArcSin[c*x])^n,x],x] /;
FreeQ[{a,b,c},x] && RationalFunctionQ[RFx,x] && IGtQ[n,0]
```

```
Int[RFx_*(a_+b_.*ArcCos[c_.*x_])^n_.,x_Symbol] :=
  Int[ExpandIntegrand[RFx*(a+b*ArcCos[c*x])^n,x],x] /;
FreeQ[{a,b,c},x] && RationalFunctionQ[RFx,x] && IGtQ[n,0]
```

2. $\int RF_x\,(d+e\,x^2)^p\,(a+b\,\text{ArcSin}[c\,x])^n\,dx$ when $n\in\mathbb{Z}^+\bigwedge c^2 d+e==0\bigwedge p-\frac{1}{2}\in\mathbb{Z}$

1: $\int RF_x\,(d+e\,x^2)^p\,\text{ArcSin}[c\,x]^n\,dx$ when $n\in\mathbb{Z}^+\bigwedge c^2 d+e==0\bigwedge p-\frac{1}{2}\in\mathbb{Z}$

Derivation: Algebraic expansion

Rule: If $n\in\mathbb{Z}^+\bigwedge c^2 d+e==0\bigwedge p-\frac{1}{2}\in\mathbb{Z}$, then

$$\int RF_x\,(d+e\,x^2)^p\,\text{ArcSin}[c\,x]^n\,dx\;\longrightarrow\;\int (d+e\,x^2)^p\,\text{ArcSin}[c\,x]^n\,\text{ExpandIntegrand}[RF_x,\,x]\,dx$$

Program code:

```
Int[RFx_*(d_+e_.*x_^2)^p_*ArcSin[c_.*x_]^n_.,x_Symbol] :=
  With[{u=ExpandIntegrand[(d+e*x^2)^p*ArcSin[c*x]^n,RFx,x]},
  Int[u,x] /;
 SumQ[u]] /;
FreeQ[{c,d,e},x] && RationalFunctionQ[RFx,x] && IGtQ[n,0] && EqQ[c^2*d+e,0] && IntegerQ[p-1/2]
```

```
Int[RFx_*(d_+e_.*x_^2)^p_*ArcCos[c_.*x_]^n_.,x_Symbol] :=
  With[{u=ExpandIntegrand[(d+e*x^2)^p*ArcCos[c*x]^n,RFx,x]},
  Int[u,x] /;
 SumQ[u]] /;
FreeQ[{c,d,e},x] && RationalFunctionQ[RFx,x] && IGtQ[n,0] && EqQ[c^2*d+e,0] && IntegerQ[p-1/2]
```

2: $\int RF_x \left(d + e x^2\right)^p (a + b \text{ArcSin}[c x])^n \, dx$ when $n \in \mathbb{Z}^+ \bigwedge c^2 d + e == 0 \bigwedge p - \frac{1}{2} \in \mathbb{Z}$

- **Derivation: Algebraic expansion**
- **Rule: If** $n \in \mathbb{Z}^+ \bigwedge c^2 d + e == 0 \bigwedge p - \frac{1}{2} \in \mathbb{Z}$, **then**

$$\int RF_x \left(d + e x^2\right)^p (a + b \text{ArcSin}[c x])^n \, dx \longrightarrow \int \left(d + e x^2\right)^p \text{ExpandIntegrand}[RF_x (a + b \text{ArcSin}[c x])^n, x] \, dx$$

- **Program code:**

```
Int[RFx_*(d_+e_.*x_^2)^p_*(a_+b_.*ArcSin[c_.*x_])^n_.,x_Symbol] :=
  Int[ExpandIntegrand[(d+e*x^2)^p,RFx*(a+b*ArcSin[c*x])^n,x],x] /;
FreeQ[{a,b,c,d,e},x] && RationalFunctionQ[RFx,x] && IGtQ[n,0] && EqQ[c^2*d+e,0] && IntegerQ[p-1/2]
```

```
Int[RFx_*(d_+e_.*x_^2)^p_*(a_+b_.*ArcCos[c_.*x_])^n_.,x_Symbol] :=
  Int[ExpandIntegrand[(d+e*x^2)^p,RFx*(a+b*ArcCos[c*x])^n,x],x] /;
FreeQ[{a,b,c,d,e},x] && RationalFunctionQ[RFx,x] && IGtQ[n,0] && EqQ[c^2*d+e,0] && IntegerQ[p-1/2]
```

U: $\int u (a + b \text{ArcSin}[c x])^n \, dx$

- **Rule:**

$$\int u (a + b \text{ArcSin}[c x])^n \, dx \longrightarrow \int u (a + b \text{ArcSin}[c x])^n \, dx$$

- **Program code:**

```
Int[u_.*(a_.+b_.*ArcSin[c_.*x_])^n_.,x_Symbol] :=
  Unintegrable[u*(a+b*ArcSin[c*x])^n,x] /;
FreeQ[{a,b,c,n},x]
```

```
Int[u_.*(a_.+b_.*ArcCos[c_.*x_])^n_.,x_Symbol] :=
  Unintegrable[u*(a+b*ArcCos[c*x])^n,x] /;
FreeQ[{a,b,c,n},x]
```